USER'S GUIDE FOR RIV2--A PACKAGE FOR ROUTING AND ACCOUNTING OF RIVER DISCHARGE FOR A MODULAR, THREE-DIMENSIONAL, FINITE-DIFFERENCE, GROUND-WATER FLOW MODEL

By Roger S. Miller

U.S. GEOLOGICAL SURVEY

Open-File Report 88-345

Albuquerque, New Mexico

1988

UNITED STATES DEPARTMENT OF THE INTERIOR

DONALD PAUL HODEL, Secretary

GEOLOGICAL SURVEY

Dallas L. Peck, Director

For additional information
write to:

District Chief
U.S. Geological Survey
Water Resources Division
Pinetree Office Park
4501 Indian School Rd. NE, Suite 200
Albuquerque, New Mexico 87110

Copies of this report can
be purchased from:

U.S. Geological Survey
Books and Open-File Reports
Federal Center, Building 810
Box 25425
Denver, Colorado 80225

CONTENTS

USER'S GUIDE FOR RIV2--A PACKAGE FOR ROUTING AND ACCOUNTING OF RIVER DISCHARGE FOR A MODULAR, THREE-DIMENSIONAL, FINITE-DIFFERENCE, GROUND-WATER FLOW MODEL

By Roger S. Miller

ABSTRACT

RIV2 is a package for the U.S. Geological Survey's modular, three-dimensional, finite-difference, ground-water flow model developed by M.G. McDonald and A.W. Harbaugh that simulates river-discharge routing. RIV2 replaces RIV1, the original river package used in the model. RIV2 preserves the basic logic of RIV1, but better represents river-discharge routing.

The main features of RIV2 are:

1. The river system is divided into reaches and simulated river discharge is routed from one reach to another in a specified sequence. Within a reach, river discharge is routed from one node to the next.

2. Inflow (river discharge) entering the upstream end of a reach can be specified.

3. More than one river can be represented at one node and rivers can cross, as when representing a siphon.

4. The quantity of leakage to or from the aquifer at a given node is proportional to the hydraulic-head difference between that specified for the river and that calculated for the aquifer. Also, the quantity of leakage to the aquifer at any node can be limited by the user and, within this limit, the maximum leakage to the aquifer is the discharge available in the river. This feature allows for the simulation of intermittent rivers and drains that have no discharge routed to their upstream reaches.

5. An accounting of river discharge is maintained.

Neither stage-discharge relations nor storage in the river or river banks is simulated.

INTRODUCTION

Model development was an integral part of the U.S. Geological Survey's Southwest Alluvial Basins Regional Aquifer-System Analysis Program in parts of Colorado, New Mexico, and Texas (Wilkins and others, 1980). The model of the Mesilla Basin in southern New Mexico needed to represent the hydraulic connection between the aquifer system and the river system. The complexity of the river system, composed of the Rio Grande, irrigation canals, and drains, required that river discharge be routed from one node to another and that an accounting of this discharge be recorded for each node.

The modular, three-dimensional, finite-difference, ground-water flow model developed by McDonald and Harbaugh (1984) has a river package, RIV1, that contains most of the essential elements needed in modeling the Mesilla Basin. However, the RIV1 package could not route the river discharge from one model node to another or maintain an accounting for the river discharge. A new river package called RIV2 was developed to add these features. RIV2, developed from RIV1, contains similar data-entry, data-output, and logical structures.

This report documents input requirements for the RIV2 package. The computations used in routing river discharge are explained as is the mathematical simulation of the river-aquifer leakage relation. Output features also are explained.

USER'S GUIDE FOR RIV2

The modeling concepts necessary for the operation of RIV2 differ little from those for RIV1. The differences are largely due to features adapted from the modeling code of Posson and others (1980) and Hearne (1982). The RIV2 code represents a number of nodes that simulate leakage from or to an overlying river. Certain features of a river that would be essential in a surface-water model, such as storage in the channel or banks, are not represented because RIV2, like RIV1, is considered to be a boundary condition in a ground-water model, not a surface-water model.

The rate of leakage at each node is directly proportional to the difference between the hydraulic head in the aquifer and the stage of the river, but is limited to the lesser of either a user-specified maximum or the discharge available in the river. This feature allows for simulation of intermittent and ephemeral rivers. Leakage from the aquifer to the river is not limited in RIV2.

The user needs to supply the hydraulic-connection coefficient, the limiting maximum rate of leakage to the aquifer, and the river stage for each node. It is possible for the user to respecify the river characteristics (stage, hydraulic-connection coefficient, and limiting maximum rate of leakage to the aquifer and river stage) for each stress period. The hydraulic-connection coefficient, CRIV, may be defined as the conductance of the reach of the riverbed with units of length squared per unit time:

$$CRIV = K'A'/b \qquad (1)$$

where K' = vertical hydraulic conductivity of the riverbed material;

A' = area of the river channel; and

b = thickness of the riverbed material.

The river discharge for a node is equal to the river discharge into the node minus the leakage to the aquifer or plus the leakage from the aquifer. The river stage, the wetted perimeter of the river channel, and the conductance of the riverbed material in a river vary with the discharge of the river. The constant values used in RIV2 limit its accuracy, but the error probably is not as great as it would be if the aquifer were allowed to gain more water from the river than the river contained.

The river-discharge-routing procedure in RIV2 uses a higher order structure that is not used in RIV1. A river, as represented in the framework of the model, consists of one or more reaches, and each reach consists of one or more nodes. (This definition of the term "reach" is distinctly different from that of RIV1.) A node may be part of more than one river reach. The river discharge at the upstream end of a reach consists of the river discharge from upstream reaches plus any user-specified tributary inflow. The river discharge from the downstream end of a reach may be routed to any downstream reach. The structure allows representation of tributaries.

RIV2, like RIV1, separates the leakage term into explicit and implicit parts. The explicit part of the leakage term is added to the variable RHS. (RHS is the right side of a finite-difference equation and is an accumulation of terms that are independent of hydraulic head at the current time step. Terms in RHS are defined by various model packages.) The term added to RHS may have either of two forms. If the hydraulic head computed for the aquifer during the previous iteration was greater than the hydraulic head required to produce the limiting value of leakage to the aquifer, then the following FORTRAN assignment is made:

```
RHS = RHS-CRIV*HRIV                                    (2)
```

where HRIV is the river stage, and other terms are as previously defined. If the hydraulic head computed for the aquifer during the previous iteration was less than or equal to the hydraulic head required to produce the limiting value of leakage to the aquifer, then the assignment is:

```
RHS = RHS-CRIV*(HRIV-HMIN)                             (3)
```

where HMIN is the hydraulic head required to produce the limiting value of leakage to the aquifer, and other terms are as previously defined.

The implicit part of the leakage term is added to the variable HCOF. (HCOF is the coefficient of hydraulic head for the node (J,I,K) in the finite-difference equation.) The implicit term may, like the explicit term, have either of two forms. If the hydraulic head computed for the aquifer during the previous iteration was greater than the hydraulic head required to produce the limiting value of leakage to the aquifer, then the following FORTRAN assignment is made:

```
HCOF = HCOF-CRIV                                       (4)
```

where all terms are as previously defined. The implicit term is zero when the hydraulic head computed for the aquifer during the previous iteration was less than or equal to the hydraulic head necessary to produce the limiting value of leakage to the aquifer. In this instance, the leakage term included in the solution algorithm is explicit.

Input Requirements

Input to the RIV2 package is read on the unit specified in IUNIT(4).

For each simulation:

1. Data: MXRIVR IRIVCB
   Format: I10 I10

2. Data: MXREAC
   Format: I10

For each stress period:

3. Data: ITMP
   Format: I10

4. Data: NR
   Format: I10

5. | Data: | NREA | NNRE | RQIN | NADD |
   |---|---|---|---|---|
   | Format: | I10 | I10 | F10.0 | I10 |

   (Item 5 consists of one record for each river reach active during the current stress period. The reaches need to be specified in downstream order.)

6. | Data: | Layer | Row | Column | Stage | Cond | Qmax |
   |---|---|---|---|---|---|---|
   | Format: | I10 | I10 | I10 | F10.0 | F10.0 | F10.0 |

   (Item 6 consists of one record for each river node active during the current stress period. The nodes need to be specified in downstream order, consistent with the specification of the river reaches.)

Explanation of fields:

MXRIVR is the maximum number of river nodes active at one time.

IRIVCB is a flag and a unit number.
- If IRIVCB > 0, then it is the unit number on which node-by-node flow terms will be recorded whenever ICBCFL (see Output Control) is set.
- If IRIVCB = 0, then node-by-node flow terms will be neither printed nor recorded.
- If IRIVCB < 0, then river leakage for each reach will be printed whenever ICBCFL is set.

MXREAC is the maximum number of river reaches active at one time.

ITMP is a flag and a counter.
    If ITMP < 0, then river data from last stress period will be re-used.
    If ITMP ≥ 0, then ITMP will be the number of river nodes active during the current stress period.

NR is the number of river reaches active in the current stress period.

NREA is the river-reach number.

NNRE is the number of river nodes in the reach.

RQIN is the river discharge added at the upstream end of the reach.

NADD is the number of the downstream reach (zero, if none).

Layer is the layer number of the river node.

Row is the row number of the river node.

Column is the column number of the river node.

Stage is the hydraulic head in the river.

Cond is the riverbed hydraulic conductance.

Qmax is the maximum allowable leakage to the aquifer.

Code Listing and Explanation

Each of the four modules of the RIV2 package is explained in narrative form and the code for each of these modules is listed. This format is similar to that used in the model documentation (McDonald and Harbaugh, 1984).

Narrative for Module RIV2AL

This module allocates space in the X array to store the river-discharge-routing information. Following are major features of this module:

1. Print message identifying the package and initialize NRIVER (number of river nodes) and NR (number of river reaches).

2. Read and print MXRIVR (the maximum number of river nodes), IRIVCB (the unit number for saving node-by-node flow terms or a flag indicating whether node-by-node flow terms should be printed), and MXREAC (the maximum number of river reaches).

3. Set LCRIVR, which will point to the first element in the river-node list (RIVR), equal to ISUM, which is currently pointing to the first unallocated element in the X array.

4. Calculate the amount of space needed for the river-node list (six values for each node--row, column, layer, stage, riverbed conductance, and maximum river leakage) and add it to ISUM.

5. Print the number of elements in the X array used by the river-node list.

6. Set LCRIVQ, which will point to the first element in the river-discharge list, equal to ISUM, which is currently pointing to the first unallocated element in the X array.

7. Calculate the amount of space needed for the river-discharge list (one value for each node), and add it to ISUM.

8. Print the number of elements in the X array used by the river-discharge list.

9. Set LCREAC, which will point to the first element in the river-reach list, equal to ISUM, which is currently pointing to the first unallocated element in the X array.

10. Calculate the amount of space needed for the river-reach list (four values for each reach--reach number, number of nodes in the reach, river discharge to be added at the upstream end of the reach, and the number of the downstream reach), and add it to ISUM.

11. Print the number of elements in the X array used by the river-reach list.

12. Print the space allocated to the river-discharge-routing package.

13. RETURN.

```
      SUBROUTINE RIV2AL(ISUM,LENX,LCRIVR,MXRIVR,NRIVER,
     1                  LCRIVQ,LCREAC,MXREAC,NR,IN,IOUT,IRIVCB)
C
C-----VERSION 1436 24MAR1984 RIV2AL
C     ************************************************************
C     ALLOCATE ARRAY STORAGE FOR RIVERS
C     ************************************************************
C
C     SPECIFICATIONS:
C     ------------------------------------------------------------
C     ------------------------------------------------------------
C
C1------IDENTIFY PACKAGE AND INITIALIZE NRIVER AND NR.
      WRITE(IOUT,1)
    1 FORMAT(1H0,'RIV2 -- RIVER PACKAGE, VERSION 2,  4/25/84')
      NRIVER=0
      NR=0
C
C
C2------READ AND PRINT MXRIVR AND IRIVCB(UNIT# FOR BUDGET)
      READ(IN,2)MXRIVR,IRIVCB
    2 FORMAT(2I10)
      WRITE(IOUT,3)MXRIVR
    3 FORMAT(1H ,'MAXIMUM OF',I5,' RIVER NODES')
      READ(IN,20)MXREAC
   20 FORMAT(I10)
      WRITE(IOUT,21)MXREAC
```

```
   21 FORMAT(1X,'MAXIMUM OF',I5,' REACHES')
      IF(IRIVCB.GT.0) WRITE(IOUT,9) IRIVCB
    9 FORMAT(1X,'CELL BUDGET WILL BE SAVED ON UNIT',I3)
C
C3------SET LCRIVR EQUAL TO ADDRESS OF FIRST UNUSED SPACE IN X.
      LCRIVR=ISUM
C
C4------CALCULATE AMOUNT OF SPACE USED BY NODE LIST.
      ISP=6*MXRIVR
      ISUM=ISUM+ISP
C
C5------PRINT AMOUNT OF SPACE USED BY NODE LIST.
      WRITE (IOUT,4)ISP
    4 FORMAT(1X,I6,
     1' ELEMENTS IN X ARRAY ARE USED FOR RIVER NODES')
C
C6------SET LCRIVQ EQUAL TO ADDRESS OF FIRST UNUSED SPACE IN X.
      LCRIVQ=ISUM
C
C7------CALCULATE AMOUNT OF SPACE USED BY RIVER FLUX.
      ISP=MXRIVR
      ISUM=ISUM+ISP
```

```
      C
C8------PRINT AMOUNT OF SPACE USED BY RIVER FLUX.
      WRITE (IOUT,27)ISP
   27 FORMAT(1X,I6,' ELEMENTS IN X ARRAY ARE USED FOR RIVER FLUX')
C
C9------SET LCREAC EQUAL TO ADDRESS OF FIRST UNUSED SPACE IN X.
      LCREAC=ISUM
C
C10-----CALCULATE AMOUNT OF SPACE USED BY REACH LIST.
      ISP=4*MXREAC
      ISUM=ISUM+ISP
C
C11-----PRINT AMOUNT OF SPACE USED BY REACH LIST.
      WRITE(IOUT,31)ISP
   31 FORMAT(1X,I6,' ELEMENTS IN X ARRAY ARE USED FOR REACHES')
C
C12-----PRINT AMOUNT OF SPACE ALLOCATED.
      ISUM1=ISUM-1
      WRITE(IOUT,5)ISUM1,LENX
    5 FORMAT(1X,I6,' ELEMENTS OF X ARRAY USED OUT OF',I7)
      IF(ISUM1.GT.LENX) WRITE(IOUT,6)
    6 FORMAT(1X,'   ***X ARRAY MUST BE DIMENSIONED LARGER***')
C
C13-----RETURN
      RETURN
      END
```

Narrative for Module RIV2RP

This module reads data to develop the river-discharge-routing lists. Following are major features of this module:

1. Read ITMP. ITMP is the number of river nodes or a flag indicating that river-discharge-routing data specified for the previous stress period will be re-used.

2. Test ITMP. If ITMP is less than zero, the river-discharge-routing data read for the last stress period will be re-used. Print a message to that effect and RETURN.

3. If ITMP is equal to or greater than zero, it is the number of river nodes for this stress period. Set the number of river nodes (NRIVER) in the current stress period equal to ITMP.

4. Compare the number of river nodes (NRIVER) in the current stress period to the number specified as the maximum for the simulation (MXRIVR). If NRIVER is greater than MXRIVR, STOP.

5. If NRIVER is less than or equal to MXRIVR, print the number of river nodes in the current stress period (NRIVER).

6. Determine if there are any river nodes. If there are no river nodes in the current stress period (NRIVER equals zero), bypass further river-discharge processing.

7. Read NR. NR is the number of river reaches defined for this stress period.

8. Compare the number of river reaches (NR) in the current stress period to the number specified as the maximum for the simulation (MXREAC). If NR is greater than MXREAC, STOP.

9. Print the number of river reaches in the current stress period (NR).

10. For each river reach, read and print the reach number, number of nodes in the reach, river discharge to be added at the upstream end of the reach, and the number of the next river reach downstream.

11. Read data on each river reach.

12. Test NADD. NADD must specify a river reach downstream from the current river reach (NADD must equal zero or NADD be greater than NREA). If it is not, print a message and STOP.

13. Test NADD. NADD needs to specify a river reach within the specified range (NADD must be less than or equal to NR). If it is not, print a message and STOP.

14. Read and print layer, row, column, river stage, riverbed conductance, and maximum river leakage for each river node. Print river-node number (nodes numbered in sequence) for each river node.

15. Print an end-of-file message and STOP.

16. Print a format error message for the river-reach list and STOP.

17. Print a format error message for the river-node list and STOP.

18. RETURN.

```
      SUBROUTINE RIV2RP(RIVR,NRIVER,REACH,NR,MXRIVR,MXREAC,IN,
     1                  IOUT)
C
C
C-----VERSION 1436 25MAR1984 RIV2RP
C     ************************************************************
C     READ RIVER REACH #, NODES PER REACH, UPSTREAM FLUX,
C     DESINATION, NODE INDICES, STAGE, CONDUCTANCE AND MAXIMUM
C     RATE OF RIVER LOSS.
C     ************************************************************
C
C     SPECIFICATIONS:
C     ------------------------------------------------------------
      DIMENSION RIVR(6,MXRIVR),REACH(4,MXREAC)
C     ------------------------------------------------------------
C
C
C1------READ ITMP(NUMBER OF RIVER NODES OR FLAG TO REUSE DATA)
      READ(IN,8)ITMP
    8 FORMAT(I10)
C
C2------TEST ITMP.
      IF(ITMP.GE.0)GO TO 50
C
```

```
C2A-----IF ITMP <0 THEN REUSE DATA FROM LAST STRESS PERIOD.
      WRITE(IOUT,7)
    7 FORMAT(1H0,'REUSING RIVER NODES FROM LAST STRESS PERIOD')
      GO TO 260
C
C3------IF ITMP ≥ ZERO THEN IT IS THE NUMBER OF RIVER NODES
   50 NRIVER=ITMP
C
C4------IF NRIVER >MXRIVR THEN STOP.
      IF(NRIVER.LE.MXRIVR)GO TO 100
      WRITE(IOUT,99)NRIVER,MXRIVR
   99 FORMAT(1H0,'NRIVER(',I4,') IS GREATER THAN MXRIVR(',I4,')')
C
C4A-----ABNORMAL STOP.
      STOP
C
C5------PRINT NUMBER OF RIVER NODES IN THIS STRESS PERIOD.
  100 WRITE(IOUT,1)NRIVER
    1 FORMAT(1H0,//1X,I5,' RIVER NODES')
C
C6------IF THERE ARE NO RIVER NODES THEN RETURN.
      IF(NRIVER.EQ.0) GO TO 260
C
C7------READ NUMBER OF RIVER REACHES.
      READ(IN,20)NR
   20 FORMAT(I10)
```

```
C
C8------IF NR  >MXREAC THEN STOP.
      IF(NR.LE.MXREAC)GO TO 101
      WRITE(IOUT,98)NR,MXREAC
   98 FORMAT(1H0,'NR(',I4,') IS GREATER THAN MXREAC(',I4,')')
C
C8A-----ABNORMAL STOP.
      STOP
C
C9------PRINT NUMBER OF RIVER REACHES.
  101 WRITE(IOUT,21)NR
   21 FORMAT(1X,'DIVIDED INTO',I5,' REACHES')
C
C10-----READ AND PRINT DATA ON RIVER REACHES.
      WRITE(IOUT,22)
   22 FORMAT(//22X,'REACH #',3X,'NODES',4X,'RQ',9X,'NEXT REACH',
     1/22X,45('-'))
C11-----READ DATA ON EACH RIVER NODE
      DO 120 II=1,NR
      READ(IN,23,END=251,ERR=252)REACH(1,II),REACH(2,II),
     1    REACH(3,II),REACH(4,II)
C-------REACH(1,II)=NREA
C-------REACH(2,II)=NNRE
C-------REACH(3,II)=RQIN
C-------REACH(4,II)=NADD
```

```
   23 FORMAT(2I10,F10.0,I10)
      WRITE(IOUT,24)REACH(1,II),REACH(2,II),REACH(3,II),
     1    REACH(4,II)
   24 FORMAT(15X,2I10,G13.4,I10)
C12-----CHECK FOR DOWNSTREAM ORDER IN SPECIFIED RIVER REACHES
      IF(REACH(4,II).NE.0.AND.REACH(4,II).LE.REACH(1,II))THEN
         WRITE(1,FMT=*)'REACHES NOT SPECIFIED IN DOWNSTREAM ORDER'
         WRITE(1,FMT=*)'(NADD MUST BE GREATER THAN NREA)'
C12A----ABNORMAL STOP
         STOP
         END IF
C13-----CHECK FOR EXISTENCE OF NEXT NODE
      IF(REACH(4,II).GT.NR)THEN
         WRITE(1,FMT=*)'DOWNSTREAM REACH DOES NOT EXIST'
         WRITE(1,FMT=*)'(NADD MUST BE LESS THAN NR)'
C13A----ABNORMAL STOP
         STOP
         END IF
  120 CONTINUE
C
C
```

```
C14-----READ AND PRINT DATA ON EACH RIVER NODE
      WRITE(IOUT,3)
    3 FORMAT(//17X,'LAYER',5X,'ROW',5X,'COL ',
     1'  STAGE          K''A/B              QMAX '
     1'          RIVER NODE'/1X,17X,80('-'))
      DO 250 II=1,NRIVER
      READ(IN,4,END=251,ERR=253)RIVR(1,II),RIVR(2,II),RIVR(3,II),
     1   RIVR(4,II),RIVR(5,II),RIVR(6,II)
    4 FORMAT(3I10,3F10.0)
C-------RIVR(1,II)=K
C-------RIVR(2,II)=I
C-------RIVR(3,II)=J
C-------RIVR(4,II)=STAGE
C-------RIVR(5,II)=COND
C-------RIVR(6,II)=QMAX
      WRITE(IOUT,5)RIVR(1,II),RIVR(2,II),RIVR(3,II),
     1   RIVR(4,II),RIVR(5,II),RIVR(6,II)
    5 FORMAT(1X,15X,I4,I9,I8,G13.4,G14.4,G19.4,I10)
  250 CONTINUE
      GO TO 260
C15-----ERROR STATEMENT FOR END OF FILE
 251  WRITE(1,FMT=*)'END OF RIVER INPUT ENCOUNTERED PREMATURELY'
      STOP
C16-----ERROR  STATEMENT FOR APPARENT FORMAT ERROR
C         IN REACH LIST
```

```
 252  WRITE(1,FMT=*)'FORMAT ERROR IN READING RIVER-REACH DATA'
      STOP
C17-----ERROR STATEMENT FOR APPARENT FORMAT ERROR
C         IN RIVER-NODE LIST
 253  WRITE(1,FMT=*)'FORMAT ERROR IN READING RIVER NODE '
     1 'PARAMETERS'
      STOP
C
C18-----RETURN
  260 RETURN
      END
```

Narrative for Module RIV2FM

This module adds terms representing river leakage to the accumulators HCOF and RHS. Following are major features of this module:

1. If NRIVER is less than or equal to zero, there are no river reaches in the current stress period. RETURN.

2. Initialize the pointers (K3 and K4) for the first river reach and set the river discharge (array Q) in this iteration to zero.

3. For each river reach defined, do steps 4-13. Start by incrementing the river discharge into the first river node in the river reach by the specified river discharge in (RQIN).

4. For each river node in the current river reach, do steps 5-10. Start by incrementing the river discharge (Q) into the river node by the computed river discharge to the river node (QIN).

5. Determine the column (IC), row (IR), and layer (IL).

6. If the river node is external [IBOUND(IC,IR,IL) is less than or equal to zero], bypass processing on this river node and go on to the next river node.

7. Because the river node is internal, set the river-node parameters [river stage, riverbed conductance, and aquifer hydraulic head that would produce leakage equal to QMAX, (RBOT1) and aquifer hydraulic head that would produce leakage equal to the river discharge to the node (RBOT2)].

8. Set HMIN to the aquifer hydraulic head (RBOT1 or RBOT2) that would produce the lesser of the limiting river leakages (specified maximum river to aquifer leakage or total river discharge).

9. If the hydraulic head (HNEW) of the aquifer is less than or equal to HMIN, add the term -CRIV*(HRIV-HMIN) to the accumulator RHS. This step limits the maximum river leakage to QMAX (the specified maximum river leakage) or to Q (the river discharge to the node), whichever is less. If the hydraulic head of the aquifer (HNEW) is greater than HMIN, add the term -CRIV*HRIV to the accumulator RHS, and the term -CRIV to the accumulator HCOF. Compute the river discharge as the difference between the river discharge to the river node and river leakage from the river node.

10. Store the remaining river discharge in QIN.

11. Find the index (IR) of the first river node in the downstream river reach.

12. If IR is not 1 (indicating that this is the last reach on the river), route the remaining river discharge (QIN) to the first river node in the downstream river reach.

13. Set pointers for the downstream river reach.

14. RETURN.

```
      SUBROUTINE RIV2FM(NRIVER,MXRIVR,RIVR,NR,MXREAC,REACH,HNEW,
     1                  HCOF,RHS,IBOUND,NCOL,NROW,NLAY,Q)
C
C-----VERSION 1436 25MAR1984 RIV2FM
C     ************************************************************
C     ADD RIVER TERMS TO RHS AND HCOF
C     COMPUTE RIVER DISCHARGES
C     ************************************************************
C
C     SPECIFICATIONS:
C     ------------------------------------------------------------
C
      DOUBLE PRECISION HNEW
      DIMENSION RIVR(6,MXRIVR),HNEW(NCOL,NROW,NLAY),
     1          HCOF(NCOL,NROW,NLAY),RHS(NCOL,NROW,NLAY),
     2          IBOUND(NCOL,NROW,NLAY),REACH(4,MXREAC),
     3          Q(MXRIVR)
C     ------------------------------------------------------------
C
C
C1------IF NRIVER ≤ 0 THERE ARE NO RIVERS. RETURN.
      IF(NRIVER.LE.0)RETURN
C
```

```
C2------INITIALIZE DO-LOOP PARAMETERS AND RESET Q
      DO 977 I=1,NRIVER
         Q(I)=0.
 977     CONTINUE
      K3=1
      K4=REACH(2,1)
      QIN=0.
C
C3------LOOP THROUGH ALL RIVER REACHES.
      DO 955 K=1,NR
C3A-----INCREMENT FLOW INTO THE REACH BY RQIN (SPECIFIED INFLOW)
         Q(K3)=Q(K3)+REACH(3,K)
         QIN=0.
C
C4------LOOP THROUGH ALL NODES WITHIN THE REACH
         DO 966 L=K3,K4
C4A-----INCREMENT RIVER FLUX IN THE NODE BY QIN (COMPUTED FLOW
C          FROM THE UPSTREAM NODE)
         Q(L)=Q(L)+QIN
C
C5------FIND NODE INDICES
            IL=RIVR(1,L)
            IR=RIVR(2,L)
            IC=RIVR(3,L)
C
C6------SKIP IF NODE IS OUTSIDE MODEL BOUNDARIES
            IF(IBOUND(IC,IR,IL).LE.0)GO TO 966
```

```
C
C7------FIND NODE PARAMETERS
            HRIV=RIVR(4,L)
            CRIV=RIVR(5,L)
            RBOT1=HRIV-(RIVR(6,L)/CRIV)
            RBOT2=HRIV-(Q(L)/CRIV)
C
C8------SET THE MAXIMUM RIVER LOSS TO QMAX OR TOTAL RIVER FLOW
C-------WHICHEVER IS SMALLEST
            HMIN=AMAX1(RBOT1,RBOT2)
            HHNEW=HNEW(IC,IR,IL)
C
C9------COMPUTE LEAKAGE TERMS AND RIVER FLOW WHEN HHNEW.LE.HMIN
            IF(HHNEW.LE.HMIN)GO TO 96
            RHS(IC,IR,IL)=RHS(IC,IR,IL)-CRIV*HRIV
            HCOF(IC,IR,IL)=HCOF(IC,IR,IL)-CRIV
            Q(L)=Q(L)-CRIV*(HRIV-HHNEW)
            GO TO 976
 96         RHS(IC,IR,IL)=RHS(IC,IR,IL)-CRIV*(HRIV-HMIN)
            Q(L)=Q(L)-CRIV*(HRIV-HMIN)
C
C10-----FLOW INTO THE DOWNSTREAM NODE IS STORED IN QIN
 976        QIN=Q(L)
 966        CONTINUE
C
```

```
C11-----FIND DOWNSTREAM REACH.
          I=REACH(4,K)
          IR=1
          DO 9555 II=1,I-1
             IR=IR+REACH(2,II)
 9555        CONTINUE
C
C12------PASS QIN TO THE FIRST NODE IN THE DOWNSTREAM REACH
          IF(IR.NE.1)THEN
             Q(IR)=Q(IR)+QIN
             END IF
C
C13------SET COUNTERS FOR NEXT REACH
          IF(K.EQ.NR)GO TO 955
          K3=K4+1
          K4=K3+REACH(2,K+1)-1
 955      CONTINUE
C14-------RETURN
       RETURN
       END
```

Narrative for Module RIV2BD

This module calculates rates and volumes of water transferred between the aquifer and rivers, and computes the remaining discharge in the rivers. Following are major features of this module:

1. Initialize the node-by-node flow-term flag (IBD), the rate accumulators (RATIN and RATOUT), and the river-discharge term (QR).

2. If there are no river reaches (NRIVER equals zero), put zeros into the budget terms for rivers and go to step 200.

3. Test to see if node-by-node flow terms are to be saved on disk. They will not be saved if either of the following conditions holds: (1) This is not the proper time step (ICBCFL equals zero), or (2) node-by-node flow terms are not to be saved for rivers during this simulation (IRIVCB is less than or equal to zero). If node-by-node flow terms will be saved for this package, set the node-by-node flow flag (IBD), set ISV to the unit number for output, and clear the buffer in which they will be accumulated (BUFF).

4. Initialize K3 and K4 to point at the first and last river nodes of the first river reach.

5. For each river reach defined, do steps 6-17. Start by incrementing the river discharge into the first river node in the river reach by the specified river discharge in (RQIN).

6. For each river node in the current river reach, do steps 7-14. Start by incrementing the river discharge (QR) into the river node by the computed river discharge to the river node (QIN).

7. Determine the row, column, and layer of the river node.

8. If the river node is external [IBOUND(I,J,K) is less than or equal to zero], bypass further processing of this river node.

9. Obtain node parameters from river-node list and define the limiting values of RBOT1 and RBOT2.

10. Set HMIN to the value of RBOT1 or RBOT2 that would produce the lesser of specified maximum river leakage or total river discharge.

11. If the river leakage is less than the maximum allowed (HHNEW greater than HMIN), set RATE equal to the conductance of the riverbed times the river stage minus the hydraulic head of the aquifer in the river node [RATE=CRIV*(HRIV-HHNEW)]. If the river leakage is greater than the maximum allowed (HHNEW is less than or equal to HMIN), set RATE equal to the maximum allowed river leakage [RATE=CRIV*(HRIV-HMIN)].

12. If budget terms for individual river nodes are to be saved, then add the RATE to the buffer (BUFF).

13. Compute the remaining river discharge and save the rate into the downstream river node in QIN.

14. Accumulate the river leakage for the mass balance. If RATE is negative, add it to RATOUT. If RATE is positive, add it to RATIN.

15. Identify the downstream river reach and locate the first river node in the river reach.

16. If the downstream river reach has been located (IR not equal to 1), route the remaining river discharge (QIN) to the first river node in the downstream river reach.

17. Reset the counters K3 and K4 for the next river reach.

18. Determine if budget terms for individual river nodes are to be saved (IBD equals 1). If they are, call module UBDSAV to record the buffer (BUFF) on the disk.

19. Move RATIN and RATOUT into the VBVL array for printing by BAS2OT. Add RATIN and RATOUT multiplied by time-step length (DELT) to the volume accumulators in VBVL for printing by BAS1OT. Move the river-budget-term labels to VBVM for printing by BAS1OT.

20. Increment the budget-term counter (MSUM).

21. RETURN.

```
      SUBROUTINE RIV2BD(NRIVER,MXRIVR,RIVR,NR,MXREAC,REACH,
     1        IBOUND,HNEW,NCOL,NROW,NLAY,DELT,VBVL,VBNM,QR,MSUM,
     2        KSTP,KPER,IRIVCB,ICBCFL,BUFF,IOUT)
C-----VERSION 1436 25MAR1984 RIV2BD
C     ************************************************************
C     CALCULATE VOLUMETRIC BUDGET FOR RIVERS
C     ************************************************************
C
C     SPECIFICATIONS:
C     ------------------------------------------------------------
      DOUBLE PRECISION HNEW
      DIMENSION RIVR(6,MXRIVR),IBOUND(NCOL,NROW,NLAY),
     1          HNEW(NCOL,NROW,NLAY),VBVL(4,20),VBNM(4,20),
     2          BUFF(NCOL,NROW,NLAY),REACH(4,MXREAC),QR(MXRIVR)
      DIMENSION TEXT(4)
      DATA TEXT(1),TEXT(2),TEXT(3) /'   R','IVER',' LEA'/
      DATA TEXT(4)/'KAGE'/
C     ------------------------------------------------------------
C
C1------INITIALIZE CELL-BY-CELL BUDGET TERM FLAG (IBD),
C1------ACCUMULATORS (RATIN AND RATOUT), AND RIVER FLOWS (QR).
      IBD=0
      RATIN=0.
      RATOUT=0.
      DO 3 I=1,NRIVER
      QR(I)=0.
    3 CONTINUE
C
```

```
C2------IF NO REACHES KEEP ZEROES IN ACCUMULATORS.
      IF(NRIVER.EQ.0)GO TO 200
C
C3------TEST TO SEE IF CELL-BY-CELL TERMS ARE NEEDED.
      IF(ICBCFL.EQ.0  .OR. IRIVCB.LE.0 ) GO TO 10
C
C3A-----CELL-BY-CELL TERMS ARE NEEDED, SET IBD, ISV (UNIT NUMBER
C--------FOR OUTPUT), AND CLEAR BUFFER
      IBD=1
      ISV=IRIVCB
      DO 5 IL=1,NLAY
      DO 5 IR=1,NROW
      DO 5 IC=1,NCOL
      BUFF(IC,IR,IL)=0.
    5 CONTINUE
C
C4------INITIALIZE POINTERS FOR THE FIRST REACH
   10 K3=1
      K4=REACH(2,1)
C
C5------LOOP THROUGH ALL RIVER REACHES.
      DO 955 K=1,NR
C5A-----INCREMENT FLOW INTO THE REACH BY THE SPECIFIED INFLOW
C-------(RQIN)
         QR(K3)=QR(K3)+REACH(3,K)
         QIN=0.
```

```
C
C6------LOOP THROUGH ALL NODES WITHIN THE REACH
        DO 966 L=K3,K4
C
C6A-----INCREMENT RIVER FLUX IN THE NODE BY QIN (COMPUTED FLOW
C         FROM THE UPSTREAM NODE)
           QR(L)=QR(L)+QIN
C
C7------FIND NODE INDICES
           IL=RIVR(1,L)
           IR=RIVR(2,L)
           IC=RIVR(3,L)
C
C8------SKIP IF NODE IS OUTSIDE MODEL BOUNDARIES
           IF(IBOUND(IC,IR,IL).LE.0)GO TO 966
C
C9------FIND NODE PARAMETERS
           HRIV=RIVR(4,L)
           CRIV=RIVR(5,L)
           RBOT1=HRIV-(RIVR(6,L)/CRIV)
           RBOT2=HRIV-(QR(L)/CRIV)
C
C10-----SET MAXIMUM RIVER LOSS TO QMAX OR TOTAL RIVER FLOW
           HMIN=AMAX1(RBOT1,RBOT2)
           HHNEW=HNEW(IC,IR,IL)
C
```

```
C11-----COMPUTE FLOWS IN OR OUT OF GROUNDWATER
            IF(HHNEW.LE.HMIN)RATE=CRIV*(HRIV-HMIN)
            IF(HHNEW.GT.HMIN)RATE=CRIV*(HRIV-HHNEW)
C
C12-----IF IBD.EQ.1, ACCUMULATE RATE IN BUFF
            IF(IBD.EQ.1)BUFF(IC,IR,IL)=BUFF(IC,IR,IL)+RATE
C
C13-----COMPUTE RIVER FLOW AND FLOW TO DOWNSTREAM NODE
            QR(L)=QR(L)-RATE
            QIN=QR(L)
C
C14-----ACCUMULATE RATES FOR MASS BALANCE
            IF(RATE)94,966,96
   94       RATOUT=RATOUT-RATE
            GO TO 966
   96       RATIN=RATIN+RATE
  966       CONTINUE
C
C15-----FIND DOWNSTREAM REACH.
         I=REACH(4,K)
         IR=1
         DO 9555 II=1,I-1
            IR=IR+REACH(2,II)
 9555       CONTINUE
```

```
C
C16-----PASS QIN TO THE FIRST NODE IN THE DOWNSTREAM REACH
         IF(IR.NE.1)THEN
            QR(IR)=QR(IR)+QIN
            END IF
C
C17-----SET COUNTERS FOR NEXT REACH
         IF(K.EQ.NR)GO TO 955
         K3=K4+1
         K4=K3+REACH(2,K+1)-1
 955     CONTINUE
C
C18-----IF BUDGET TERMS WILL BE SAVED CALL UBUDSV TO WRITE THEM ON
C-------DISK
      IF(IBD.EQ.1) CALL UBUDSV(KSTP,KPER,TEXT,IRIVCB,BUFF,NCOL,
     1                          NROW,NLAY,ISV)
C
C19-----MOVE RATES,VOLUMES & LABELS INTO ARRAYS FOR PRINTING.
  200 VBVL(3,MSUM)=RATIN
      VBVL(4,MSUM)=RATOUT
      VBVL(1,MSUM)=VBVL(1,MSUM)+RATIN*DELT
      VBVL(2,MSUM)=VBVL(2,MSUM)+RATOUT*DELT
      VBNM(1,MSUM)=TEXT(1)
      VBNM(2,MSUM)=TEXT(2)
      VBNM(3,MSUM)=TEXT(3)
      VBNM(4,MSUM)=TEXT(4)
C
```

```
C20-----INCREMENT BUDGET TERM COUNTER
      MSUM=MSUM+1
C
C21-----RETURN
      RETURN
      END
```

REFERENCES CITED

Hearne, G.A., 1982, Supplement to the New Mexico three-dimensional model (Supplement to Open-File Report 80-421): U.S. Geological Survey Open-File Report 82-857, 90 p.

McDonald, M.G., and Harbaugh, A.W., 1984, A modular three-dimensional finite-difference ground-water flow model: U.S. Geological Survey Open-File Report 83-875, 528 p.

Posson, D.R., Hearne, G.A., Tracy, J.V., and Frenzel, P.F., 1980, A computer program for simulating geohydrologic systems in three dimensions: U.S. Geological Survey Open-File Report 80-421, 795 p.

Wilkins, D.W., Scott, W.B., and Kaehler, C.A., 1980, Planning report for the Southwest Alluvial Basins (east) Regional Aquifer-System Analysis, parts of Colorado, New Mexico, and Texas: U.S. Geological Survey Open-File Report 80-564, 39 p.

☆ U.S. GOVERNMENT PRINTING OFFICE: 1988—673-471/5,004 REGION NO. 8